Rules for normalizing algebraic functions

1. $\int u\,(a+b\,(c\,x^n)^q)^p\,dx$ when $n\,q\in\mathbb{Z}$

1: $\int (a+b\,(c\,x^n)^q)^p\,dx$ when $n\,q\in\mathbb{Z}$

- **Derivation: Integration by substitution and piecewise constant extraction**
- **Basis:** $f\left[(c\,x^n)^{1/n}\right] == \frac{x}{(c\,x^n)^{1/n}}\,f\left[(c\,x^n)^{1/n}\right]\,\partial_x\,(c\,x^n)^{1/n}$
- **Basis:** $\partial_x\,\frac{x}{(c\,x^n)^{1/n}} == 0$
- **Note: This rule should be generalized to handle arbitrary functions of** $(c\,x^n)^{1/n}$.
- **Rule 1.5.4.1.1: If** $n\,q\in\mathbb{Z}$, **then**

$$\int (a+b\,(c\,x^n)^q)^p\,dx \longrightarrow \frac{x}{(c\,x^n)^{1/n}}\,\text{Subst}\left[\int (a+b\,x^{n\,q})^p\,dx,\ x,\ (c\,x^n)^{1/n}\right]$$

- **Program code:**

```
Int[(a_.+b_.*(c_.*x_^n_)^q_)^p_.,x_Symbol] :=
  x/(c*x^n)^(1/n)*Subst[Int[(a+b*x^(n*q))^p,x],x,(c*x^n)^(1/n)] /;
FreeQ[{a,b,c,q,n,p},x] && IntegerQ[n*q]
```

2: $\int x^m\,(a+b\,(c\,x^n)^q)^p\,dx$ when $n\,q\in\mathbb{Z}\,\land\,m\in\mathbb{Z}$

- **Derivation: Integration by substitution and piecewise constant extraction**
- **Basis: If** $m\in\mathbb{Z}$, **then** $x^m\,f\left[(c\,x^n)^{1/n}\right] == \frac{x^{m+1}}{(c\,x^n)^{(m+1)/n}}\,\left((c\,x^n)^{1/n}\right)^m\,f\left[(c\,x^n)^{1/n}\right]\,\partial_x\,(c\,x^n)^{1/n}$
- **Basis:** $\partial_x\,\frac{x^{m+1}}{(c\,x^n)^{(m+1)/n}} == 0$
- **Note: This rule should be generalized to handle arbitrary functions of** $(c\,x^n)^{1/n}$ **times** x^m.
- **Rule 1.5.4.1.2: If** $n\,q\in\mathbb{Z}\,\land\,m\in\mathbb{Z}$, **then**

$$\int x^m\,(a+b\,(c\,x^n)^q)^p\,dx \longrightarrow \frac{x^{m+1}}{(c\,x^n)^{(m+1)/n}}\,\text{Subst}\left[\int x^m\,(a+b\,x^{n\,q})^p\,dx,\ x,\ (c\,x^n)^{1/n}\right]$$

- **Program code:**

```
Int[x_^m_.*(a_.+b_.*(c_.*x_^n_)^q_)^p_.,x_Symbol] :=
  x^(m+1)/(c*x^n)^((m+1)/n)*Subst[Int[x^m*(a+b*x^(n*q))^p,x],x,(c*x^n)^(1/n)] /;
FreeQ[{a,b,c,m,n,p,q},x] && IntegerQ[n*q] && IntegerQ[m]
```

2: $\int x^m\,(e\,(a+b\,x^n)^r)^p\,(f\,(c+d\,x^n)^s)^q\,dx$

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_x \frac{(e\,(a+b\,x^n)^r)^p\,(f\,(c+d\,x^n)^s)^q}{(a+b\,x^n)^{p\,r}\,(c+d\,x^n)^{q\,s}} == 0$
- **Rule 1.5.4.2:**

$$\int x^m\,(e\,(a+b\,x^n)^r)^p\,(f\,(c+d\,x^n)^s)^q\,dx \longrightarrow \frac{(e\,(a+b\,x^n)^r)^p\,(f\,(c+d\,x^n)^s)^q}{(a+b\,x^n)^{p\,r}\,(c+d\,x^n)^{q\,s}}\int x^m\,(a+b\,x^n)^{p\,r}\,(c+d\,x^n)^{q\,s}\,dx$$

- **Program code:**

```
Int[x_^m_.*(e_.*(a_+b_.*x_^n_.)^r_.)^p_*(f_.*(c_+d_.*x_^n_.)^s_)^q_,x_Symbol] :=
  (e*(a+b*x^n)^r)^p*(f*(c+d*x^n)^s)^q/((a+b*x^n)^(p*r)*(c+d*x^n)^(q*s))*
    Int[x^m*(a+b*x^n)^(p*r)*(c+d*x^n)^(q*s),x] /;
FreeQ[{a,b,c,d,e,f,m,n,p,q,r,s},x]
```

3. $\int u\left(e\,\frac{a+b\,x^n}{c+d\,x^n}\right)^p dx$

1: $\int u\left(e\,\frac{a+b\,x^n}{c+d\,x^n}\right)^p dx$ when $b\,c-a\,d==0$

- **Derivation: Algebraic simplification**
- **Basis: If** $b\,c-a\,d==0$, **then** $\frac{a+b\,z}{c+d\,z}==\frac{b}{d}$
- **Rule 1.5.4.3.1: If** $b\,c-a\,d==0$, **then**

$$\int u\left(e\,\frac{a+b\,x^n}{c+d\,x^n}\right)^p dx \longrightarrow \left(\frac{b\,e}{d}\right)^p\int u\,dx$$

- **Program code:**

```
Int[u_.*(e_.*(a_.+b_.*x_^n_.)/(c_+d_.*x_^n_.))^p_,x_Symbol] :=
  (b*e/d)^p*Int[u,x] /;
FreeQ[{a,b,c,d,e,n,p},x] && EqQ[b*c-a*d,0]
```

2. $\int u\left(e\,\frac{a+b\,x^n}{c+d\,x^n}\right)^p dx$ when $b\,c-a\,d\neq 0$

1: $\int u \left(e \frac{a+b x^n}{c+d x^n}\right)^p dx$ when $b d e > 0 \bigwedge c < \frac{a d}{b}$

- **Derivation: Algebraic simplification**
- **Basis: If** $b d e > 0 \bigwedge \frac{a d}{b} \le c$, **then** $\left(e \frac{a+b z}{c+d z}\right)^p == \frac{(e (a+b z))^p}{(c+d z)^p}$
- **Rule 1.5.4.3.2.1: If** $b d e > 0 \bigwedge c < \frac{a d}{b}$, **then**

$$\int u \left(e \frac{a+b x^n}{c+d x^n}\right)^p dx \longrightarrow \int \frac{u (e (a+b x^n))^p}{(c+d x^n)^p} dx$$

- **Program code:**

```
Int[u_.*(e_.*(a_.+b_.*x_^n_.)/(c_+d_.*x_^n_.))^p_,x_Symbol] :=
  Int[u*(e*(a+b*x^n))^p/(c+d*x^n)^p,x] /;
FreeQ[{a,b,c,d,e,n,p},x] && GtQ[b*d*e,0] && GtQ[c-a*d/b,0]
```

2. $\int u \left(e \frac{a+b x^n}{c+d x^n}\right)^p dx$ when $\neg \left(b d e > 0 \bigwedge \frac{a d}{b} \le c\right)$

1: $\int \left(e \frac{a+b x^n}{c+d x^n}\right)^p dx$ when $\frac{1}{n} \in \mathbb{Z}$

- **Derivation: Integration by substitution**
- **Basis: If** $\frac{1}{n} \in \mathbb{Z} \bigwedge q \in \mathbb{Z}^+$, **then** $\left(e \frac{a+b x^n}{c+d x^n}\right)^p == \frac{q e (b c-a d)}{n} \text{Subst}\left[\frac{x^{q (p+1)-1} (-a e+c x^q)^{\frac{1}{n}-1}}{(b e-d x^q)^{\frac{1}{n}+1}}, x, \left(e \frac{a+b x^n}{c+d x^n}\right)^{1/q}\right] \partial_x \left(e \frac{a+b x^n}{c+d x^n}\right)^{1/q}$
- **Rule 1.5.4.3.2.2.1: If** $\frac{1}{n} \in \mathbb{Z}$, **let** $q = \text{Denominator}[p]$, **then**

$$\int \left(e \frac{a+b x^n}{c+d x^n}\right)^p dx \longrightarrow \frac{q e (b c-a d)}{n} \text{Subst}\left[\int \frac{x^{q (p+1)-1} (-a e+c x^q)^{\frac{1}{n}-1}}{(b e-d x^q)^{\frac{1}{n}+1}} dx, x, \left(e \frac{a+b x^n}{c+d x^n}\right)^{1/q}\right]$$

- **Program code:**

```
Int[(e_.*(a_.+b_.*x_^n_.)/(c_+d_.*x_^n_.))^p_,x_Symbol] :=
  With[{q=Denominator[p]},
  q*e*(b*c-a*d)/n*Subst[
    Int[x^(q*(p+1)-1)*(-a*e+c*x^q)^(1/n-1)/(b*e-d*x^q)^(1/n+1),x],x,(e*(a+b*x^n)/(c+d*x^n))^(1/q)]] /;
FreeQ[{a,b,c,d,e},x] && FractionQ[p] && IntegerQ[1/n]
```

2: $\int x^m \left(e\,\frac{a+b\,x^n}{c+d\,x^n}\right)^p dx$ when $\frac{m+1}{n}\in\mathbb{Z}$

- **Derivation: Integration by substitution**

- **Basis: If** $\frac{m+1}{n}\in\mathbb{Z}\bigwedge q\in\mathbb{Z}^+$, **then** $x^m\left(e\,\frac{a+b\,x^n}{c+d\,x^n}\right)^p == \frac{q\,e\,(b\,c-a\,d)}{n}\,\text{Subst}\left[\frac{x^{q\,(p+1)-1}\,(-a\,e+c\,x^q)^{\frac{m+1}{n}-1}}{(b\,e-d\,x^q)^{\frac{m+1}{n}+1}},\,x,\,\left(e\,\frac{a+b\,x^n}{c+d\,x^n}\right)^{1/q}\right]\partial_x\left(e\,\frac{a+b\,x^n}{c+d\,x^n}\right)^{1/q}$

- **Rule 1.5.4.3.2.2.2: If** $\frac{m+1}{n}\in\mathbb{Z}$, **let** $q=\text{Denominator}[p]$, **then**

$$\int x^m\left(e\,\frac{a+b\,x^n}{c+d\,x^n}\right)^p dx \longrightarrow \frac{q\,e\,(b\,c-a\,d)}{n}\,\text{Subst}\left[\int\frac{x^{q\,(p+1)-1}\,(-a\,e+c\,x^q)^{\frac{m+1}{n}-1}}{(b\,e-d\,x^q)^{\frac{m+1}{n}+1}}\,dx,\,x,\,\left(e\,\frac{a+b\,x^n}{c+d\,x^n}\right)^{1/q}\right]$$

- **Program code:**

```
Int[x_^m_.*(e_.*(a_.+b_.*x_^n_.)/(c_+d_.*x_^n_.))^p_,x_Symbol] :=
  With[{q=Denominator[p]},
  q*e*(b*c-a*d)/n*Subst[
    Int[x^(q*(p+1)-1)*(-a*e+c*x^q)^(Simplify[(m+1)/n]-1)/(b*e-d*x^q)^(Simplify[(m+1)/n]+1),x],x,(e*(a+b*x^n)/(c+d*x^n))^(1/q)]] /;
FreeQ[{a,b,c,d,e,m,n},x] && FractionQ[p] && IntegerQ[Simplify[(m+1)/n]]
```

3: $\int P_x^r\left(e\,\frac{a+b\,x^n}{c+d\,x^n}\right)^p dx$ when $\frac{1}{n}\in\mathbb{Z}\bigwedge r\in\mathbb{Z}$

- **Derivation: Integration by substitution**

- **Basis: If** $\frac{1}{n}\in\mathbb{Z}\bigwedge q\in\mathbb{Z}^+$, **then** $F[x]\left(e\,\frac{a+b\,x^n}{c+d\,x^n}\right)^p == \frac{q\,e\,(b\,c-a\,d)}{n}\,\text{Subst}\left[\frac{x^{q\,(p+1)-1}\,(-a\,e+c\,x^q)^{\frac{1}{n}-1}}{(b\,e-d\,x^q)^{\frac{1}{n}+1}}\,F\left[\frac{(-a\,e+c\,x^q)^{\frac{1}{n}}}{(b\,e-d\,x^q)^{\frac{1}{n}}}\right],\,x,\,\left(e\,\frac{a+b\,x^n}{c+d\,x^n}\right)^{1/q}\right]\partial_x\left(e\,\frac{a+b\,x^n}{c+d\,x^n}\right)^{1/q}$

- **Rule 1.5.4.3.2.2.3: If** $\frac{1}{n}\in\mathbb{Z}$, **let** $q=\text{Denominator}[p]$, **then**

$$\int P_x^r\left(e\,\frac{a+b\,x^n}{c+d\,x^n}\right)^p dx \longrightarrow \frac{q\,e\,(b\,c-a\,d)}{n}\,\text{Subst}\left[\int\frac{x^{q\,(p+1)-1}\,(-a\,e+c\,x^q)^{\frac{1}{n}-1}}{(b\,e-d\,x^q)^{\frac{1}{n}+1}}\,\text{Subst}\left[P_x,\,x,\,\frac{(-a\,e+c\,x^q)^{\frac{1}{n}}}{(b\,e-d\,x^q)^{\frac{1}{n}}}\right]^r dx,\,x,\,\left(e\,\frac{a+b\,x^n}{c+d\,x^n}\right)^{1/q}\right]$$

- **Program code:**

```
Int[u_^r_.*(e_.*(a_.+b_.*x_^n_.)/(c_+d_.*x_^n_.))^p_,x_Symbol] :=
  With[{q=Denominator[p]},
  q*e*(b*c-a*d)/n*Subst[Int[SimplifyIntegrand[x^(q*(p+1)-1)*(-a*e+c*x^q)^(1/n-1)/(b*e-d*x^q)^(1/n+1)*
    ReplaceAll[u,x→(-a*e+c*x^q)^(1/n)/(b*e-d*x^q)^(1/n)]^r,x],x],x,(e*(a+b*x^n)/(c+d*x^n))^(1/q)]] /;
FreeQ[{a,b,c,d,e},x] && PolynomialQ[u,x] && FractionQ[p] && IntegerQ[1/n] && IntegerQ[r]
```

4: $\int x^m P_x^r \left(e \frac{a+b x^n}{c+d x^n}\right)^p dx$ when $\frac{1}{n} \in \mathbb{Z} \bigwedge (m \mid r) \in \mathbb{Z}$

- **Derivation: Integration by substitution**
- **Basis: If** $\frac{1}{n} \in \mathbb{Z} \bigwedge m \in \mathbb{Z} \bigwedge q \in \mathbb{Z}^+$**, then**

$$x^m F[x] \left(e \frac{a+b x^n}{c+d x^n}\right)^p = \frac{q e (b c-a d)}{n} \text{Subst}\left[\frac{x^{q (p+1)-1} (-a e+c x^q)^{\frac{m+1}{n}-1}}{(b e-d x^q)^{\frac{m+1}{n}+1}} F\left[\frac{(-a e+c x^q)^{\frac{1}{n}}}{(b e-d x^q)^{\frac{1}{n}}}\right], x, \left(e \frac{a+b x^n}{c+d x^n}\right)^{1/q}\right] \partial_x \left(e \frac{a+b x^n}{c+d x^n}\right)^{1/q}$$

- **Rule 1.5.4.3.2.2.4: If** $\frac{1}{n} \in \mathbb{Z} \bigwedge (m \mid r) \in \mathbb{Z}$**, let** $q = \text{Denominator}[p]$**, then**

$$\int x^m P_x^r \left(e \frac{a+b x^n}{c+d x^n}\right)^p dx \to \frac{q e (b c-a d)}{n} \text{Subst}\left[\int \frac{x^{q (p+1)-1} (-a e+c x^q)^{\frac{m+1}{n}-1}}{(b e-d x^q)^{\frac{m+1}{n}+1}} \text{Subst}\left[P_x, x, \frac{(-a e+c x^q)^{\frac{1}{n}}}{(b e-d x^q)^{\frac{1}{n}}}\right]^r dx, x, \left(e \frac{a+b x^n}{c+d x^n}\right)^{1/q}\right]$$

- **Program code:**

```
Int[x_^m_.*u_^r_.*(e_.*(a_.+b_.*x_^n_.)/(c_+d_.*x_^n_.))^p_,x_Symbol] :=
  With[{q=Denominator[p]},
  q*e*(b*c-a*d)/n*Subst[Int[SimplifyIntegrand[x^(q*(p+1)-1)*(-a*e+c*x^q)^((m+1)/n-1)/(b*e-d*x^q)^((m+1)/n+1)*
    ReplaceAll[u,x→(-a*e+c*x^q)^(1/n)/(b*e-d*x^q)^(1/n)]^r,x],x],x,(e*(a+b*x^n)/(c+d*x^n))^(1/q)]] /;
FreeQ[{a,b,c,d,e},x] && PolynomialQ[u,x] && FractionQ[p] && IntegerQ[1/n] && IntegersQ[m,r]
```

4. $\int u \left(a+b \left(\frac{c}{x}\right)^n\right)^p dx$

1: $\int \left(a+b \left(\frac{c}{x}\right)^n\right)^p dx$

- **Derivation: Integration by substitution**
- **Basis:** $F\left[\frac{c}{x}\right] = -c \,\text{Subst}\left[\frac{F[x]}{x^2}, x, \frac{c}{x}\right] \partial_x \frac{c}{x}$
- **Rule 1.5.4.4.1:**

$$\int \left(a+b \left(\frac{c}{x}\right)^n\right)^p dx \to -c \,\text{Subst}\left[\int \frac{(a+b x^n)^p}{x^2} dx, x, \frac{c}{x}\right]$$

- **Program code:**

```
Int[(a_.+b_.*(c_./x_)^n_)^p_,x_Symbol] :=
  -c*Subst[Int[(a+b*x^n)^p/x^2,x],x,c/x] /;
FreeQ[{a,b,c,n,p},x]
```

2. $\int (d\,x)^m \left(a+b\left(\frac{c}{x}\right)^n\right)^p dx$

1: $\int x^m \left(a+b\left(\frac{c}{x}\right)^n\right)^p dx$ when $m\in\mathbb{Z}$

- **Derivation: Integration by substitution**
- **Basis: If** $m\in\mathbb{Z}$, **then** $x^m F\left[\frac{c}{x}\right] = -c^{m+1}\,\mathrm{Subst}\left[\frac{F[x]}{x^{m+2}},\,x,\,\frac{c}{x}\right]\partial_x\frac{c}{x}$
- **Rule 1.5.4.4.2.1: If** $m\in\mathbb{Z}$, **then**

$$\int x^m \left(a+b\left(\frac{c}{x}\right)^n\right)^p dx \longrightarrow -c^{m+1}\,\mathrm{Subst}\left[\int\frac{(a+b\,x^n)^p}{x^{m+2}}dx,\,x,\,\frac{c}{x}\right]$$

- **Program code:**

```
Int[x_^m_.*(a_.+b_.*(c_./x_)^n_)^p_.,x_Symbol] :=
  -c^(m+1)*Subst[Int[(a+b*x^n)^p/x^(m+2),x],x,c/x] /;
FreeQ[{a,b,c,n,p},x] && IntegerQ[m]
```

2: $\int (d\,x)^m \left(a+b\left(\frac{c}{x}\right)^n\right)^p dx$ when $m\notin\mathbb{Z}$

- **Derivation: Piecewise constant extraction and integration by substitution**
- **Basis:** $\partial_x\left((d\,x)^m\left(\frac{c}{x}\right)^m\right) = 0$
- **Basis:** $F\left[\frac{c}{x}\right] = -c\,\mathrm{Subst}\left[\frac{F[x]}{x^2},\,x,\,\frac{c}{x}\right]\partial_x\frac{c}{x}$
- **Rule 1.5.4.4.2.2: If** $m\notin\mathbb{Z}$, **then**

$$\int (d\,x)^m \left(a+b\left(\frac{c}{x}\right)^n\right)^p dx \longrightarrow (d\,x)^m\left(\frac{c}{x}\right)^m\int\frac{\left(a+b\left(\frac{c}{x}\right)^n\right)^p}{\left(\frac{c}{x}\right)^m}dx \longrightarrow -c\,(d\,x)^m\left(\frac{c}{x}\right)^m\mathrm{Subst}\left[\int\frac{(a+b\,x^n)^p}{x^{m+2}}dx,\,x,\,\frac{c}{x}\right]$$

- **Program code:**

```
Int[(d_.*x_)^m_*(a_.+b_.*(c_./x_)^n_)^p_.,x_Symbol] :=
  -c*(d*x)^m*(c/x)^m*Subst[Int[(a+b*x^n)^p/x^(m+2),x],x,c/x] /;
FreeQ[{a,b,c,d,m,n,p},x] && Not[IntegerQ[m]]
```

5. $\int u\left(a+b\left(\frac{d}{x}\right)^n+c\left(\frac{d}{x}\right)^{2n}\right)^p dx$

1: $\int\left(a+b\left(\frac{d}{x}\right)^n+c\left(\frac{d}{x}\right)^{2n}\right)^p dx$

- **Derivation: Integration by substitution**
- **Basis:** $F\left[\frac{d}{x}\right] == -d\,\text{Subst}\left[\frac{F[x]}{x^2},\ x,\ \frac{d}{x}\right]\partial_x\frac{d}{x}$
- **Rule 1.5.4.5.1:**

$$\int\left(a+b\left(\frac{d}{x}\right)^n+c\left(\frac{d}{x}\right)^{2n}\right)^p dx \longrightarrow -d\,\text{Subst}\left[\int\frac{\left(a+b\,x^n+c\,x^{2n}\right)^p}{x^2}dx,\ x,\ \frac{d}{x}\right]$$

- **Program code:**

```
Int[(a_.+b_.*(d_./x_)^n_+c_.*(d_./x_)^n2_.)^p_.,x_Symbol] :=
  -d*Subst[Int[(a+b*x^n+c*x^(2*n))^p/x^2,x],x,d/x] /;
FreeQ[{a,b,c,d,n,p},x] && EqQ[n2,2*n]
```

2. $\int(e\,x)^m\left(a+b\left(\frac{d}{x}\right)^n+c\left(\frac{d}{x}\right)^{2n}\right)^p dx$

1: $\int x^m\left(a+b\left(\frac{d}{x}\right)^n+c\left(\frac{d}{x}\right)^{2n}\right)^p dx$ when $m\in\mathbb{Z}$

- **Derivation: Integration by substitution**
- **Basis: If** $m\in\mathbb{Z}$**, then** $x^m F\left[\frac{d}{x}\right] == -d^{m+1}\,\text{Subst}\left[\frac{F[x]}{x^{m+2}},\ x,\ \frac{d}{x}\right]\partial_x\frac{d}{x}$
- **Rule 1.5.4.5.2.1: If** $m\in\mathbb{Z}$**, then**

$$\int x^m\left(a+b\left(\frac{d}{x}\right)^n+c\left(\frac{d}{x}\right)^{2n}\right)^p dx \longrightarrow -d^{m+1}\,\text{Subst}\left[\int\frac{\left(a+b\,x^n+c\,x^{2n}\right)^p}{x^{m+2}}dx,\ x,\ \frac{d}{x}\right]$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*(d_./x_)^n_+c_.*(d_./x_)^n2_.)^p_.,x_Symbol] :=
  -d^(m+1)*Subst[Int[(a+b*x^n+c*x^(2*n))^p/x^(m+2),x],x,d/x] /;
FreeQ[{a,b,c,d,n,p},x] && EqQ[n2,2*n] && IntegerQ[m]
```

2: $\int (e\,x)^m \left(a+b\left(\frac{d}{x}\right)^n+c\left(\frac{d}{x}\right)^{2n}\right)^p dx$ when $m\notin\mathbb{Z}$

- **Derivation: Piecewise constant extraction and integration by substitution**
- **Basis:** $\partial_x\left((e\,x)^m\left(\frac{d}{x}\right)^m\right)=0$
- **Basis:** $F\left[\frac{d}{x}\right]=-d\,\text{Subst}\left[\frac{F[x]}{x^2},\,x,\,\frac{d}{x}\right]\partial_x\frac{d}{x}$
- **Rule 1.5.4.5.2.2: If** $m\notin\mathbb{Z}$**, then**

$$\int (e\,x)^m\left(a+b\left(\frac{d}{x}\right)^n+c\left(\frac{d}{x}\right)^{2n}\right)^p dx \to (e\,x)^m\left(\frac{d}{x}\right)^m\int\frac{\left(a+b\left(\frac{d}{x}\right)^n+c\left(\frac{d}{x}\right)^{2n}\right)^p}{\left(\frac{d}{x}\right)^m}dx \to -d\,(e\,x)^m\left(\frac{d}{x}\right)^m\text{Subst}\left[\int\frac{\left(a+b\,x^n+c\,x^{2n}\right)^p}{x^{m+2}}dx,\,x,\,\frac{d}{x}\right]$$

- **Program code:**

```
Int[(e_.*x_)^m_*(a_+b_.*(d_./x_)^n_+c_.*(d_./x_)^n2_.)^p_.,x_Symbol] :=
  -d*(e*x)^m*(d/x)^m*Subst[Int[(a+b*x^n+c*x^(2*n))^p/x^(m+2),x],x,d/x] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && EqQ[n2,2*n] && Not[IntegerQ[m]]
```

6. $\int u\left(a+b\left(\frac{d}{x}\right)^n+c\,x^{-2n}\right)^p dx$ when $2n\in\mathbb{Z}$

1: $\int \left(a+b\left(\frac{d}{x}\right)^n+c\,x^{-2n}\right)^p dx$ when $2n\in\mathbb{Z}$

- **Derivation: Integration by substitution**
- **Basis:** $F\left[\frac{d}{x}\right]=-d\,\text{Subst}\left[\frac{F[x]}{x^2},\,x,\,\frac{d}{x}\right]\partial_x\frac{d}{x}$
- **Rule 1.5.4.6.1: If** $2n\in\mathbb{Z}$**, then**

$$\int\left(a+b\left(\frac{d}{x}\right)^n+c\,x^{-2n}\right)^p dx \to \int\left(a+b\left(\frac{d}{x}\right)^n+\frac{c}{d^{2n}}\left(\frac{d}{x}\right)^{2n}\right)^p dx \to -d\,\text{Subst}\left[\int\frac{\left(a+b\,x^n+\frac{c}{d^{2n}}x^{2n}\right)^p}{x^2}dx,\,x,\,\frac{d}{x}\right]$$

- **Program code:**

```
Int[(a_.+b_.*(d_./x_)^n_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  -d*Subst[Int[(a+b*x^n+c/d^(2*n)*x^(2*n))^p/x^2,x],x,d/x] /;
FreeQ[{a,b,c,d,n,p},x] && EqQ[n2,-2*n] && IntegerQ[2*n]
```

2. $\int (e\,x)^m \left(a+b\left(\frac{d}{x}\right)^n+c\,x^{-2\,n}\right)^p dx$ when $2\,n\in\mathbb{Z}$

1: $\int x^m \left(a+b\left(\frac{d}{x}\right)^n+c\,x^{-2\,n}\right)^p dx$ when $2\,n\in\mathbb{Z}\,\land\,m\in\mathbb{Z}$

- **Derivation: Integration by substitution**
- **Basis: If** $m\in\mathbb{Z}$, **then** $x^m\,F\left[\frac{d}{x}\right] = -d^{m+1}\,\text{Subst}\left[\frac{F[x]}{x^{m+2}},\,x,\,\frac{d}{x}\right]\partial_x\frac{d}{x}$
- **Rule 1.5.4.6.2.1: If** $2\,n\in\mathbb{Z}\,\land\,m\in\mathbb{Z}$, **then**

$$\int x^m\left(a+b\left(\frac{d}{x}\right)^n+c\,x^{-2\,n}\right)^p dx\;\longrightarrow\;\int x^m\left(a+b\left(\frac{d}{x}\right)^n+\frac{c}{d^{2\,n}}\left(\frac{d}{x}\right)^{2\,n}\right)^p dx\;\longrightarrow\;-d^{m+1}\,\text{Subst}\left[\int\frac{\left(a+b\,x^n+\frac{c}{d^{2\,n}}\,x^{2\,n}\right)^p}{x^{m+2}}\,dx,\,x,\,\frac{d}{x}\right]$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*(d_./x_)^n_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  -d^(m+1)*Subst[Int[(a+b*x^n+c/d^(2*n)*x^(2*n))^p/x^(m+2),x],x,d/x] /;
FreeQ[{a,b,c,d,n,p},x] && EqQ[n2,-2*n] && IntegerQ[2*n] && IntegerQ[m]
```

2: $\int (e\,x)^m \left(a+b\left(\frac{d}{x}\right)^n+c\,x^{-2\,n}\right)^p dx$ when $2\,n\in\mathbb{Z}\,\land\,m\notin\mathbb{Z}$

- **Derivation: Piecewise constant extraction and integration by substitution**
- **Basis:** $\partial_x\left((e\,x)^m\left(\frac{d}{x}\right)^m\right) = 0$
- **Basis:** $F\left[\frac{d}{x}\right] = -d\,\text{Subst}\left[\frac{F[x]}{x^2},\,x,\,\frac{d}{x}\right]\partial_x\frac{d}{x}$
- **Rule 1.5.4.6.2.2: If** $2\,n\in\mathbb{Z}\,\land\,m\notin\mathbb{Z}$, **then**

$$\int (e\,x)^m\left(a+b\left(\frac{d}{x}\right)^n+c\,x^{-2\,n}\right)^p dx\;\longrightarrow\;(e\,x)^m\left(\frac{d}{x}\right)^m\int\frac{\left(a+b\left(\frac{d}{x}\right)^n+\frac{c}{d^{2\,n}}\left(\frac{d}{x}\right)^{2\,n}\right)^p}{\left(\frac{d}{x}\right)^m}\,dx\;\longrightarrow\;-d\,(e\,x)^m\left(\frac{d}{x}\right)^m\text{Subst}\left[\int\frac{\left(a+b\,x^n+\frac{c}{d^{2\,n}}\,x^{2\,n}\right)^p}{x^{m+2}}\,dx,\,x,\,\frac{d}{x}\right]$$

- **Program code:**

```
Int[(e_.*x_)^m_*(a_+b_.*(d_./x_)^n_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  -d*(e*x)^m*(d/x)^m*Subst[Int[(a+b*x^n+c/d^(2*n)*x^(2*n))^p/x^(m+2),x],x,d/x] /;
FreeQ[{a,b,c,d,e,n,p},x] && EqQ[n2,-2*n] && Not[IntegerQ[m]] && IntegerQ[2*n]
```

7. Binomial products

1. Linear

1: $\int u^m \, dx$ when $u == a + b x$

- **Derivation: Algebraic normalization**
- **Rule: If $u == a + b x$, then**

$$\int u^m \, dx \longrightarrow \int (a + b x)^m \, dx$$

- **Program code:**

```
Int[u_^m_,x_Symbol] :=
  Int[ExpandToSum[u,x]^m,x] /;
FreeQ[m,x] && LinearQ[u,x] && Not[LinearMatchQ[u,x]]
```

2: $\int u^m v^n \, dx$ when $u == a + b x \wedge v == c + d x$

- **Derivation: Algebraic normalization**
- **Rule: If $u == a + b x \wedge v == c + d x$, then**

$$\int u^m v^n \, dx \longrightarrow \int (a + b x)^m (c + d x)^n \, dx$$

- **Program code:**

```
Int[u_^m_.*v_^n_.,x_Symbol] :=
  Int[ExpandToSum[u,x]^m*ExpandToSum[v,x]^n,x] /;
FreeQ[{m,n},x] && LinearQ[{u,v},x] && Not[LinearMatchQ[{u,v},x]]
```

3: $\int u^m v^n w^p \, dx$ when $u == a + b x \land v == c + d x \land w == e + f x$

- **Derivation: Algebraic normalization**
- **Rule: If** $u == a + b x \land v == c + d x \land w == e + f x$**, then**

$$\int u^m v^n w^p \, dx \longrightarrow \int (a + b x)^m (c + d x)^n (e + f x)^p \, dx$$

- **Program code:**

```
Int[u_^m_.*v_^n_.*w_^p_.,x_Symbol] :=
  Int[ExpandToSum[u,x]^m*ExpandToSum[v,x]^n*ExpandToSum[w,x]^p,x] /;
FreeQ[{m,n,p},x] && LinearQ[{u,v,w},x] && Not[LinearMatchQ[{u,v,w},x]]
```

4: $\int u^m v^n w^p z^q \, dx$ when $u == a + b x \land v == c + d x \land w == e + f x \land z == g + h x$

- **Derivation: Algebraic normalization**
- **Rule: If** $u == a + b x \land v == c + d x \land w == e + f x \land z == g + h x$**, then**

$$\int u^m v^n w^p z^q \, dx \longrightarrow \int (a + b x)^m (c + d x)^n (e + f x)^p (g + h x)^q \, dx$$

- **Program code:**

```
Int[u_^m_.*v_^n_.*w_^p_.*z_^q_.,x_Symbol] :=
  Int[ExpandToSum[u,x]^m*ExpandToSum[v,x]^n*ExpandToSum[w,x]^p*ExpandToSum[z,x]^q,x] /;
FreeQ[{m,n,p,q},x] && LinearQ[{u,v,w,z},x] && Not[LinearMatchQ[{u,v,w,z},x]]
```

3. General

1: $\int u^p \, dx$ when $u == a + b x^n$

- **Derivation: Algebraic normalization**
- **Rule: If $u == a + b x^n$, then**

$$\int u^p \, dx \longrightarrow \int (a + b x^n)^p \, dx$$

- **Program code:**

```
Int[u_^p_,x_Symbol] :=
  Int[ExpandToSum[u,x]^p,x] /;
FreeQ[p,x] && BinomialQ[u,x] && Not[BinomialMatchQ[u,x]]
```

2: $\int (c x)^m u^p \, dx$ when $u == a + b x^n$

- **Derivation: Algebraic normalization**
- **Rule: If $u == a + b x^n$, then**

$$\int (c x)^m u^p \, dx \longrightarrow \int (c x)^m (a + b x^n)^p \, dx$$

- **Program code:**

```
Int[(c_.*x_)^m_.*u_^p_.,x_Symbol] :=
  Int[(c*x)^m*ExpandToSum[u,x]^p,x] /;
FreeQ[{c,m,p},x] && BinomialQ[u,x] && Not[BinomialMatchQ[u,x]]
```

3: $\int u^p v^q \, dx$ when $u == a + b x^n \wedge v == c + d x^n$

- **Derivation: Algebraic normalization**
- **Rule: If** $u == a + b x^n \wedge v == c + d x^n$**, then**

$$\int u^p v^q \, dx \longrightarrow \int (a + b x^n)^p (c + d x^n)^q \, dx$$

- **Program code:**

```
Int[u_^p_.*v_^q_.,x_Symbol] :=
  Int[ExpandToSum[u,x]^p*ExpandToSum[v,x]^q,x] /;
FreeQ[{p,q},x] && BinomialQ[{u,v},x] && EqQ[BinomialDegree[u,x]-BinomialDegree[v,x],0] && Not[BinomialMatchQ[{u,v},x]]
```

4: $\int (e x)^m u^p v^q \, dx$ when $u == a + b x^n \wedge v == c + d x^n$

- **Derivation: Algebraic normalization**
- **Rule: If** $u == a + b x^n \wedge v == c + d x^n$**, then**

$$\int (e x)^m u^p v^q \, dx \longrightarrow \int (e x)^m (a + b x^n)^p (c + d x^n)^q \, dx$$

- **Program code:**

```
Int[(e_.*x_)^m_.*u_^p_.*v_^q_.,x_Symbol] :=
  Int[(e*x)^m*ExpandToSum[u,x]^p*ExpandToSum[v,x]^q,x] /;
FreeQ[{e,m,p,q},x] && BinomialQ[{u,v},x] && EqQ[BinomialDegree[u,x]-BinomialDegree[v,x],0] && Not[BinomialMatchQ[{u,v},x]]
```

5: $\int u^m v^p w^q \, dx$ when $u == a + b x^n \bigwedge v == c + d x^n \bigwedge w == e + f x^n$

- **Derivation: Algebraic normalization**
- **Rule: If** $u == a + b x^n \bigwedge v == c + d x^n \bigwedge w == e + f x^n$**, then**

$$\int u^m v^p w^q \, dx \longrightarrow \int (a + b x^n)^m (c + d x^n)^p (e + f x^n)^q \, dx$$

- **Program code:**

```
Int[u_^m_.*v_^p_.*w_^q_.,x_Symbol] :=
  Int[ExpandToSum[u,x]^m*ExpandToSum[v,x]^p*ExpandToSum[w,x]^q,x] /;
FreeQ[{m,p,q},x] && BinomialQ[{u,v,w},x] && EqQ[BinomialDegree[u,x]-BinomialDegree[v,x],0] &&
  EqQ[BinomialDegree[u,x]-BinomialDegree[w,x],0] && Not[BinomialMatchQ[{u,v,w},x]]
```

6: $\int (g x)^m u^p v^q z^r \, dx$ when $u == a + b x^n \bigwedge v == c + d x^n \bigwedge z == e + f x^n$

- **Derivation: Algebraic normalization**
- **Rule: If** $u == a + b x^n \bigwedge v == c + d x^n \bigwedge z == e + f x^n$**, then**

$$\int (g x)^m u^p v^q z^r \, dx \longrightarrow \int (g x)^m (a + b x^n)^p (c + d x^n)^q (e + f x^n)^r \, dx$$

- **Program code:**

```
Int[(g_.*x_)^m_.*u_^p_.*v_^q_.*z_^r_.,x_Symbol] :=
  Int[(g*x)^m*ExpandToSum[u,x]^p*ExpandToSum[v,x]^q*ExpandToSum[z,x]^r,x] /;
FreeQ[{g,m,p,q,r},x] && BinomialQ[{u,v,z},x] && EqQ[BinomialDegree[u,x]-BinomialDegree[v,x],0] &&
  EqQ[BinomialDegree[u,x]-BinomialDegree[z,x],0] && Not[BinomialMatchQ[{u,v,z},x]]
```

7: $\int (c\,x)^m\,P_q[x]\,u^p\,dx$ when $u == a + b\,x^n$

- **Derivation: Algebraic normalization**
- **Rule: If** $u == a + b\,x^n$**, then**

$$\int (c\,x)^m\,P_q[x]\,u^p\,dx \longrightarrow \int (c\,x)^m\,P_q[x]\,(a + b\,x^n)^p\,dx$$

- **Program code:**

```
Int[(c_.*x_)^m_.*Pq_*u_^p_.,x_Symbol] :=
  Int[(c*x)^m*Pq*ExpandToSum[u,x]^p,x] /;
FreeQ[{c,m,p},x] && PolyQ[Pq,x] && BinomialQ[u,x] && Not[BinomialMatchQ[u,x]]
```

4. Improper

1: $\int u^p\,dx$ when $u == a\,x^j + b\,x^n$

- **Derivation: Algebraic normalization**
- **Rule: If** $u == a\,x^j + b\,x^n$**, then**

$$\int u^p\,dx \longrightarrow \int \left(a\,x^j + b\,x^n\right)^p\,dx$$

- **Program code:**

```
Int[u_^p_,x_Symbol] :=
  Int[ExpandToSum[u,x]^p,x] /;
FreeQ[p,x] && GeneralizedBinomialQ[u,x] && Not[GeneralizedBinomialMatchQ[u,x]]
```

2: $\int (c\,x)^m\,u^p\,dx$ when $u == a\,x^j + b\,x^n$

- **Derivation: Algebraic normalization**
- **Rule: If $u == a\,x^j + b\,x^n$, then**

$$\int (c\,x)^m\,u^p\,dx \longrightarrow \int (c\,x)^m\,\left(a\,x^j + b\,x^n\right)^p\,dx$$

- **Program code:**

```
Int[(c_.*x_)^m_.*u_^p_.,x_Symbol] :=
  Int[(c*x)^m*ExpandToSum[u,x]^p,x] /;
FreeQ[{c,m,p},x] && GeneralizedBinomialQ[u,x] && Not[GeneralizedBinomialMatchQ[u,x]]
```

8 Trinomial products

1. Quadratic

1: $\int u^p\,dx$ when $u == a + b\,x + c\,x^2$

- **Derivation: Algebraic normalization**
- **Rule: If $u == a + b\,x + c\,x^2$, then**

$$\int u^p\,dx \longrightarrow \int \left(a + b\,x + c\,x^2\right)^p\,dx$$

- **Program code:**

```
Int[u_^p_,x_Symbol] :=
  Int[ExpandToSum[u,x]^p,x] /;
FreeQ[p,x] && QuadraticQ[u,x] && Not[QuadraticMatchQ[u,x]]
```

2: $\int u^m v^p \, dx$ when $u == d + e x \wedge v == a + b x + c x^2$

- **Derivation: Algebraic normalization**
- **Rule: If** $u == d + e x \wedge v == a + b x + c x^2$**, then**

$$\int u^m v^p \, dx \longrightarrow \int (d + e x)^m \left(a + b x + c x^2\right)^p dx$$

- **Program code:**

```
Int[u_^m_.*v_^p_.,x_Symbol] :=
  Int[ExpandToSum[u,x]^m*ExpandToSum[v,x]^p,x] /;
FreeQ[{m,p},x] && LinearQ[u,x] && QuadraticQ[v,x] && Not[LinearMatchQ[u,x] && QuadraticMatchQ[v,x]]
```

3: $\int u^m v^n w^p \, dx$ when $u == d + e x \wedge v == f + g x \wedge w == a + b x + c x^2$

- **Derivation: Algebraic normalization**
- **Rule: If** $u == d + e x \wedge v == f + g x \wedge w == a + b x + c x^2$**, then**

$$\int u^m v^n w^p \, dx \longrightarrow \int (d + e x)^m (f + g x)^n \left(a + b x + c x^2\right)^p dx$$

- **Program code:**

```
Int[u_^m_.*v_^n_.*w_^p_.,x_Symbol] :=
  Int[ExpandToSum[u,x]^m*ExpandToSum[v,x]^n*ExpandToSum[w,x]^p,x] /;
FreeQ[{m,n,p},x] && LinearQ[{u,v},x] && QuadraticQ[w,x] && Not[LinearMatchQ[{u,v},x] && QuadraticMatchQ[w,x]]
```

4: $\int u^p\, v^q\, dx$ when $u == a + b\,x + c\,x^2 \,\land\, v == d + e\,x + f\,x^2$

- **Derivation: Algebraic normalization**
- **Rule: If $u == a + b\,x + c\,x^2 \,\land\, v == d + e\,x + f\,x^2$, then**

$$\int u^p\, v^q\, dx \,\longrightarrow\, \int \left(a + b\,x + c\,x^2\right)^p \left(d + e\,x + f\,x^2\right)^q dx$$

- **Program code:**

```
Int[u_^p_.*v_^q_.,x_Symbol] :=
  Int[ExpandToSum[u,x]^p*ExpandToSum[v,x]^q,x] /;
FreeQ[{p,q},x] && QuadraticQ[{u,v},x] && Not[QuadraticMatchQ[{u,v},x]]
```

5: $\int z^m\, u^p\, v^q\, dx$ when $z == g + h\,x \,\land\, u == a + b\,x + c\,x^2 \,\land\, v == d + e\,x + f\,x^2$

- **Derivation: Algebraic normalization**
- **Note: This normalization needs to be done before trying polynomial integration rules.**
- **Rule 1.2.1.5.N: If $z == g + h\,x \,\land\, u == a + b\,x + c\,x^2 \,\land\, v == d + e\,x + f\,x^2$, then**

$$\int z^m\, u^p\, v^q\, dx \,\longrightarrow\, \int (g + h\,x)^m \left(a + b\,x + c\,x^2\right)^p \left(d + e\,x + f\,x^2\right)^q dx$$

- **Program code:**

```
Int[z_^m_.*u_^p_.*v_^q_.,x_Symbol] :=
  Int[ExpandToSum[z,x]^m*ExpandToSum[u,x]^p*ExpandToSum[v,x]^q,x] /;
FreeQ[{m,p,q},x] && LinearQ[z,x] && QuadraticQ[{u,v},x] && Not[LinearMatchQ[z,x] && QuadraticMatchQ[{u,v},x]]
```

6: $\int P_q[x]\, u^p\, dx$ when $u == a + b x + c x^2$

- **Derivation: Algebraic normalization**
- **Rule: If $u == a + b x + c x^2$, then**

$$\int P_q[x]\, u^p\, dx \longrightarrow \int P_q[x] \left(a + b x + c x^2\right)^p dx$$

- **Program code:**

```
Int[Pq_*u_^p_.,x_Symbol] :=
  Int[Pq*ExpandToSum[u,x]^p,x] /;
FreeQ[p,x] && PolyQ[Pq,x] && QuadraticQ[u,x] && Not[QuadraticMatchQ[u,x]]
```

7: $\int u^m\, P_q[x]\, v^p\, dx$ when $u == d + e x \wedge v == a + b x + c x^2$

- **Derivation: Algebraic normalization**
- **Rule: If $u == d + e x \wedge v == a + b x + c x^2$, then**

$$\int u^m\, P_q[x]\, v^p\, dx \longrightarrow \int (d + e x)^m\, P_q[x] \left(a + b x + c x^2\right)^p dx$$

- **Program code:**

```
Int[u_^m_.*Pq_*v_^p_.,x_Symbol] :=
  Int[ExpandToSum[u,x]^m*Pq*ExpandToSum[v,x]^p,x] /;
FreeQ[{m,p},x] && PolyQ[Pq,x] && LinearQ[u,x] && QuadraticQ[v,x] && Not[LinearMatchQ[u,x] && QuadraticMatchQ[v,x]]
```

3. General

1: $\int u^p \, dx$ when $u == a + b x^n + c x^{2n}$

- **Derivation: Algebraic normalization**
- **Rule: If $u == a + b x^n + c x^{2n}$, then**

$$\int u^p \, dx \longrightarrow \int \left(a + b x^n + c x^{2n}\right)^p \, dx$$

- **Program code:**

```
Int[u_^p_,x_Symbol] :=
  Int[ExpandToSum[u,x]^p,x] /;
FreeQ[p,x] && TrinomialQ[u,x] && Not[TrinomialMatchQ[u,x]]
```

2: $\int (d x)^m u^p \, dx$ when $u == a + b x^n + c x^{2n}$

- **Derivation: Algebraic normalization**
- **Rule: If $u == a + b x^n + c x^{2n}$, then**

$$\int (d x)^m u^p \, dx \longrightarrow \int (d x)^m \left(a + b x^n + c x^{2n}\right)^p \, dx$$

- **Program code:**

```
Int[(d_.*x_)^m_.*u_^p_.,x_Symbol] :=
  Int[(d*x)^m*ExpandToSum[u,x]^p,x] /;
FreeQ[{d,m,p},x] && TrinomialQ[u,x] && Not[TrinomialMatchQ[u,x]]
```

3: $\int u^q v^p \, dx$ when $u == d + e x^n \wedge v == a + b x^n + c x^{2n}$

- **Derivation: Algebraic normalization**
- **Rule: If** $u == d + e x^n \wedge v == a + b x^n + c x^{2n}$**, then**

$$\int u^q v^p \, dx \longrightarrow \int (d + e x^n)^q \left(a + b x^n + c x^{2n}\right)^p dx$$

- **Program code:**

```
Int[u_^q_.*v_^p_.,x_Symbol] :=
  Int[ExpandToSum[u,x]^q*ExpandToSum[v,x]^p,x] /;
FreeQ[{p,q},x] && BinomialQ[u,x] && TrinomialQ[v,x] && Not[BinomialMatchQ[u,x] && TrinomialMatchQ[v,x]]
```

```
Int[u_^q_.*v_^p_.,x_Symbol] :=
  Int[ExpandToSum[u,x]^q*ExpandToSum[v,x]^p,x] /;
FreeQ[{p,q},x] && BinomialQ[u,x] && BinomialQ[v,x] && Not[BinomialMatchQ[u,x] && BinomialMatchQ[v,x]]
```

4: $\int (f x)^m z^q u^p \, dx$ when $z == d + e x^n \wedge u == a + b x^n + c x^{2n}$

- **Derivation: Algebraic normalization**
- **Rule: If** $z == d + e x^n \wedge u == a + b x^n + c x^{2n}$**, then**

$$\int (f x)^m z^q u^p \, dx \longrightarrow \int (f x)^m (d + e x^n)^q \left(a + b x^n + c x^{2n}\right)^p dx$$

- **Program code:**

```
Int[(f_.*x_)^m_.*z_^q_.*u_^p_.,x_Symbol] :=
  Int[(f*x)^m*ExpandToSum[z,x]^q*ExpandToSum[u,x]^p,x] /;
FreeQ[{f,m,p,q},x] && BinomialQ[z,x] && TrinomialQ[u,x] && Not[BinomialMatchQ[z,x] && TrinomialMatchQ[u,x]]
```

```
Int[(f_.*x_)^m_.*z_^q_.*u_^p_.,x_Symbol] :=
  Int[(f*x)^m*ExpandToSum[z,x]^q*ExpandToSum[u,x]^p,x] /;
FreeQ[{f,m,p,q},x] && BinomialQ[z,x] && BinomialQ[u,x] && Not[BinomialMatchQ[z,x] && BinomialMatchQ[u,x]]
```

5: $\int P_q[x]\,u^p\,dx$ when $u == a+b\,x^n+c\,x^{2\,n}$

- **Derivation: Algebraic normalization**
- **Rule: If $u == a+b\,x^n+c\,x^{2\,n}$, then**

$$\int P_q[x]\,u^p\,dx \longrightarrow \int P_q[x]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$$

- **Program code:**

```
Int[Pq_*u_^p_.,x_Symbol] :=
  Int[Pq*ExpandToSum[u,x]^p,x] /;
FreeQ[p,x] && PolyQ[Pq,x] && TrinomialQ[u,x] && Not[TrinomialMatchQ[u,x]]
```

6: $\int (d\,x)^m\,P_q[x]\,u^p\,dx$ when $u == a+b\,x^n+c\,x^{2\,n}$

- **Derivation: Algebraic normalization**
- **Rule: If $u == a+b\,x^n+c\,x^{2\,n}$, then**

$$\int (d\,x)^m\,P_q[x]\,u^p\,dx \longrightarrow \int (d\,x)^m\,P_q[x]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$$

- **Program code:**

```
Int[(d_.*x_)^m_.*Pq_*u_^p_.,x_Symbol] :=
  Int[(d*x)^m*Pq*ExpandToSum[u,x]^p,x] /;
FreeQ[{d,m,p},x] && PolyQ[Pq,x] && TrinomialQ[u,x] && Not[TrinomialMatchQ[u,x]]
```

4. Improper

1: $\int u^p \, dx$ when $u == a x^q + b x^n + c x^{2n-q}$

- **Derivation: Algebraic normalization**
- **Rule: If $u == a x^q + b x^n + c x^{2n-q}$, then**

$$\int u^p \, dx \longrightarrow \int \left(a x^q + b x^n + c x^{2n-q}\right)^p dx$$

- **Program code:**

```
Int[u_^p_,x_Symbol] :=
  Int[ExpandToSum[u,x]^p,x] /;
FreeQ[p,x] && GeneralizedTrinomialQ[u,x] && Not[GeneralizedTrinomialMatchQ[u,x]]
```

2: $\int (d x)^m u^p \, dx$ when $u == a x^q + b x^n + c x^{2n-q}$

- **Derivation: Algebraic normalization**
- **Rule: If $u == a x^q + b x^n + c x^{2n-q}$, then**

$$\int (d x)^m u^p \, dx \longrightarrow \int (d x)^m \left(a x^q + b x^n + c x^{2n-q}\right)^p dx$$

- **Program code:**

```
Int[(d_.*x_)^m_.*u_^p_.,x_Symbol] :=
  Int[(d*x)^m*ExpandToSum[u,x]^p,x] /;
FreeQ[{d,m,p},x] && GeneralizedTrinomialQ[u,x] && Not[GeneralizedTrinomialMatchQ[u,x]]
```

3: $\int z\,u^p\,dx$ when $z == A + B\,x^{n-q} \land u == a\,x^q + b\,x^n + c\,x^{2\,n-q}$

- **Derivation: Algebraic normalization**
- **Rule: If** $z == A + B\,x^{n-q} \land u == a\,x^q + b\,x^n + c\,x^{2\,n-q}$, **then**

$$\int z\,u^p\,dx \longrightarrow \int \left(A + B\,x^{n-q}\right) \left(a\,x^q + b\,x^n + c\,x^{2\,n-q}\right)^p dx$$

- **Program code:**

```
Int[z_*u_^p_.,x_Symbol] :=
  Int[ExpandToSum[z,x]*ExpandToSum[u,x]^p,x] /;
FreeQ[p,x] && BinomialQ[z,x] && GeneralizedTrinomialQ[u,x] &&
  EqQ[BinomialDegree[z,x]-GeneralizedTrinomialDegree[u,x],0] && Not[BinomialMatchQ[z,x] && GeneralizedTrinomialMatchQ[u,x]]
```

4: $\int (f\,x)^m\,z\,u^p\,dx$ when $z == A + B\,x^{n-q} \land u == a\,x^q + b\,x^n + c\,x^{2\,n-q}$

- **Derivation: Algebraic normalization**
- **Rule: If** $z == A + B\,x^{n-q} \land u == a\,x^q + b\,x^n + c\,x^{2\,n-q}$, **then**

$$\int (f\,x)^m\,z\,u^p\,dx \longrightarrow \int (f\,x)^m \left(A + B\,x^{n-q}\right) \left(a\,x^q + b\,x^n + c\,x^{2\,n-q}\right)^p dx$$

- **Program code:**

```
Int[(f_.*x_)^m_.*z_*u_^p_.,x_Symbol] :=
  Int[(f*x)^m*ExpandToSum[z,x]*ExpandToSum[u,x]^p,x] /;
FreeQ[{f,m,p},x] && BinomialQ[z,x] && GeneralizedTrinomialQ[u,x] &&
  EqQ[BinomialDegree[z,x]-GeneralizedTrinomialDegree[u,x],0] && Not[BinomialMatchQ[z,x] && GeneralizedTrinomialMatchQ[u,x]]
```